
Test::Builder2

TB2 CAT IS WATCHING YOU RUN TESTS



There's a lot of Testing modules on CPAN



use Test;

Back in the old days, there was Test.pm
  by Joshua Pritikin
And it was limited
And it wasn't well documented
And people didn't use it



use Test::More;

Then in early 2001 I released Test::More
And it was way better



use Test::More tests => 3;

ok  ( $thing, "thing is true" );
is  ( 1 + 1, 2 );
like( $thing, qr/stuff/ );

It was well documented
And unambiguous
And did a lot of things you want to do in tests
But you couldn't add to it
And you couldn't use it with other testing modules



1..3
ok 1
ok 2
ok 3

This is because of the test counter.



use Test::More;
use Test::Web qw(url_ok);

plan tests => 2;

is( 1 + 1, 2 );
url_ok( "http://instantrimshot.com" );

If you tried to use two different test modules together
They'd each have their own counter



1..2
ok 1
ok 1

And this would happen
which is a failure



use Test::More;

sub file_contents_ok {
    my($file1, $file2) = @_;
    open my $fh1, $file1;
    open my $fh2, $file2;

    is join("", <$fh1>),
       join("", <$fh2>),
       "compare $file1 & $file2";
}

And people wanted to write their own test functions
They were writing wrappers around Test::More functions.
This is a problem because...



not ok 23 - compare foo & bar
#  Failed test at Test/Totally/Dont/Care.pm
#    line 20938.

It would report the failure as happening where file_contents_ok() calls is()
Not at the point where you called file_contents_ok()
Which is where you care about.



Finally, people wanted detailed control over how the tests behaved
Where they output to
If they used numbers
Stuff that didn't really belong in the Test::More interface.



Test::Builder

So Test::Builder came into being



use Test::More tests => 3;
use Test::Deep qw(cmp_deeply);
use Test::Exception qw(throws_ok);

is( 1+1, 2 );

cmp_deeply( \%hash, { foo => 42 } );

throws_ok {
    $obj->method;
} qr/\Qmethod() missing arguments/;

Any Test module that uses Test::Builder
Shares the counter



1..3
ok 1 - is
ok 2 - cmp_deeply
ok 3 - throws_ok

So they can coordinate



use Test::More;

sub file_contents_ok {
    my($file1, $file2) = @_;
    open my $fh1, $file1;
    open my $fh2, $file2;

    my $builder = Test::More->builder;

    return $builder->is_eq(
       join("", <$fh1>),
       join("", <$fh2>),
       "compare $file1 & $file2"
    );
}

and you can write your own test functions



not ok 23 - compare foo & bar
#  Failed test at mytest.t
#    line 12.

that report at the right spot



$builder->use_numbers(0);
$builder->output($fh);

And you can tweak details about how the test works
While still hiding those controls from most users who don't care about it



Test::Builder
Sep 2001

9 years old

chromatic and I wrote it in 2001
Testing has come a long way in 9 years.
And Test::Builder is getting in the way of what people want to do.
There's design problems
Like what?



Die on fail.
When a test fails, stop the test.
This is how the rest of the testing world works.
Some people really want it.
I think its a bad idea, but part of my job is to get the hell out of the way.



sub is {
    my($have, $want, $name) = @_;
    
    my $builder = Test::Builder->new;
    my $ok = $builder->ok(
        $have eq $want,
        $name
    );
    
    $builder->diag(<<'DIAGNOSTIC') if !$ok;
    have: $have
    want: $want
DIAGNOSTIC

    return $ok;
}

Here's an example of how you'd might write is() in Test::Builder.
And its a pretty typical pattern for writing asserts with Test::Builder.
Assume ok() dies on failure.
Who sees the problem?



not ok 23 - foo bar
#   Failed test at ohgod.t line 42.
#     have: foo
#     want: bar

The other issue is parsable diagnostics
You explicitly cannot parse the above.
  Its a comment, the format will change.
If you want to know the file or line it failed on
Or display the differences between two long strings



not ok 23 - foo bar
    ---
    have: foo
    want: bar
    file: ohgod.t
    line: 42
    ...

We've wanted structured failure diagnostics for years
But Test::Builder can't support them



sub is {
    my($have, $want, $name) = @_;
    
    my $builder = Test::Builder->new;

    # This supplies file and line diagnostics
    my $ok = $builder->ok(
        $have eq $want,
        $name
    );
    
    # This supplies custom diagnostics
    $builder->diagnostics(
        have => $have
        want => $want
    );

    return $ok;
}

You have ok() supplying standard information like file and line
And then the assert author can add whatever they want
So when does Test::Builder print the diagnostics?



When does
 an assert end?

Turns out these all have a common problem.
Test::Builder doesn't know when an assert ends.
Because they're stacked.
is() is stacked on top of ok().
And something might be stacked on top of is().
  And so on.
Test::Builder has to do annoying bookkeeping to track the top of the stack
And there's no trigger to let it run code when the top is reached.



1..3
ok 1
ok 2
not ok 3

Then there's the output.
Test::Builder only outputs TAP.



<testsuites>
  <testsuite
   failures="0"
   errors="0"
   tests="1"
   name="_Users_schwern_tmp_ok_t"
  >
    <testcase name="1"></testcase>
    <system-out><![CDATA[1..1
ok 1
]]></system-out>
    <system-err></system-err>
  </testsuite>
</testsuites>

You might want to output JUnit to integrate with other tools
Like Hudson



TAP version 16
1..3
awesome 1
awesome 2
bogus 3

TB1 can't even support other versions of TAP
Because the TAP output is all done inline



Separate
the

formatter
TB1 needs a way to talk to different formatters



use Test::More;
use Test::NoWarnings;

is( this(), "that", "this()");

done_testing;

Finally, testing modules want to hook into the test
Test::NoWarnings adds another test at the end
Checks that there's no warnings



ok 1 - this()
ok 2 - no warnings
1..2

So you should wind up with this

TNW changes plan,
And add a test at the end



ok 1 - this()
1..1
ok 2 - no warnings
# Looks like you planned 1 test
# but ran 2.

You wind up with this.
Because the plan can be set and output in several places and in different ways.
And there's no way to TNW to know them all.



Event
Hooks

Test::Builder needs a way for 3rd party modules to hook into test events
And possibly change them



Test::Builder2

It's time to write Test::Builder2
Its in a new namespace because its incompatible with the TB1 API
But that's ok, they'll work together.



The Perl Foundation gave me a grant to do this...
two years ago
and I think I said I'd have it done in a few months
...running a bit behind schedule
Its been very start stop
    hit a design problem and have to ruminate on it
But I dedicated September to the problem



It has evolved far past what it was when I started
And for a while was out of control
But now all the major pieces are finally in place
Even if they don't all exist...



http://github.com/schwern/test-more/

There is code.
Its in the Test-Builder2 branch.
There was an alpha release, but that's a bit out of date.



One of the first decisions I made was to use a real OO system
  because I'm sick of not being able to take advantage of modern Perl
  and having its features available would prove interesting
  even if I didn't yet know what I wanted to use
So I wrote TB2 with Mouse (a small Moose)



cp -r Mouse/ Test/Builder2/Mouse/

s{Mouse}{Test::Builder2::Mouse}g;

Early on Mouse would change and break TB2
  sometimes bugs
  usually because TB2 was doing it wrong
To mitigate this, TB2 ships with its own renamed copy of Mouse.
Mouse has no dependencies.



Mouse::Tiny

    Mouse in one file



Mousse

    make_mousse 'Test::Builder2::Mouse'



package My::Test;

use Test::Builder2::Mouse;

has stuff
    is => 'rw';
has things
    is => 'rw';

And its available for test library authors to use.



$ time perl -wle 'use Moose'

real 0m0.242s
user 0m0.209s
sys  0m0.023s

$ time perl -wle 'use Mouse'

real 0m0.040s
user 0m0.029s
sys  0m0.008s

Why not use Moose?
Its too damn slow.
And it has lots of dependencies.



                     .-------.
                     | foo.t |
                     '-------'
                         |
                         |
     .-------------.    / \    .----------------.
     | Test::More  |<---   --->| Test::Whatever |
     '-------------'           '----------------'
            |                           |
            |                           |
            |                           |
            |     .---------------.     |
            '---->| Test::Builder |<----'
                  '---------------'
                          |
                          v
                       .-----.
                       | TAP |
                       '-----'
                          |
                          v
                  .---------------.
                  | Test::Harness |
                  '---------------'

Here's the flow of an assert through Test::Builder
TB1 is monolithic
Asserts, formatting, everything happens inside it
There's no further break down into other objects
It outputs only TAP, and its all ad hoc
It can't even handle other versions of TAP



                                  .-------.
                 .----------------| foo.t |----------------.
                 |                '-------'                |
                 |                    |                    |
                 |                    |                    |
                 v                    v                    v
          .------------.     .----------------.     .------------------.
          | Test::More |     | Test::Whatever |     | Test::NotUpdated |
          '------------'     '----------------'     '------------------'
                 |                |                       |
                 |                v                       v
                 |       .----------------.       .---------------.
                 '------>| Test::Builder2 |       | Test::Builder |
                         '----------------'       '---------------'
                                      |                  |
                                      v                  |
                               .-------------.           |
                               | TB2::Result |<----------'
                               '-------------'
                                      |
                                      v
        .--------------.    .-----------------------.    
        | TB2::History |<---| TB2::EventCoordinator |
        '--------------'    '-----------------------'
                                      |
    .--------------------------.      |       .---------------------.
    | TB2::Formatter::TAP::v13 |<-----'------>| TB2::Formatter::GUI |
    '--------------------------'              '---------------------'
                  |                                      |
                  v                                      |
  .-------------------------------.                      |
  | TB2::Formatter::Streamer::TAP |                      |
  '-------------------------------'                      |
                  |                                      |
                  v                                      |
               .-----.                                   |
               | TAP |                                   |
               '-----'                                   |
                  |                                      |
                  v                                      v
          .---------------.                     .-----------------.
          | Test::Harness |                     | Pretty Pictures |
          '---------------'                     '-----------------'

Test::Builder2 is a little more involved...
TB1 continues to work alongside TB2
We'll see more about that in a moment
Formatters are now their own thing
Can be swapped out
But really it makes life easier.



sub ok {
    my( $self, $test, $name ) = @_;

    if ( $self->{Child_Name} and not $self->{In_Destroy} ) {
        $name = 'unnamed test' unless defined $name;
        $self->is_passing(0);
        $self->croak("Cannot run test ($name) with active children");
    }
    # $test might contain an object which we don't want to accidentally
    # store, so we turn it into a boolean.
    $test = $test ? 1 : 0;

    lock $self->{Curr_Test};
    $self->{Curr_Test}++;

    # In case $name is a string overloaded object, force it to stringify.
    $self->_unoverload_str( \$name );

    $self->diag(<<"ERR") if defined $name and $name =~ /^[\d\s]+$/;
    You named your test '$name'.  You shouldn't use numbers for your test names.
    Very confusing.
ERR

    # Capture the value of $TODO for the rest of this ok() call
    # so it can more easily be found by other routines.
    my $todo    = $self->todo();
    my $in_todo = $self->in_todo;
    local $self->{Todo} = $todo if $in_todo;

    $self->_unoverload_str( \$todo );

    my $out;
    my $result = &share( {} );

    unless($test) {
        $out .= "not ";
        @$result{ 'ok', 'actual_ok' } = ( ( $self->in_todo ? 1 : 0 ), 0 );
    }
    else {
        @$result{ 'ok', 'actual_ok' } = ( 1, $test );
    }

    $out .= "ok";
    $out .= " $self->{Curr_Test}" if $self->use_numbers;

So here's the first half of Test::Builder->ok



    if( defined $name ) {
        $name =~ s|#|\\#|g;    # # in a name can confuse Test::Harness.
        $out .= " - $name";
        $result->{name} = $name;
    }
    else {
        $result->{name} = '';
    }

    if( $self->in_todo ) {
        $out .= " # TODO $todo";
        $result->{reason} = $todo;
        $result->{type}   = 'todo';
    }
    else {
        $result->{reason} = '';
        $result->{type}   = '';
    }

    $self->{Test_Results}[ $self->{Curr_Test} - 1 ] = $result;
    $out .= "\n";

    $self->_print($out);

    unless($test) {
        my $msg = $self->in_todo ? "Failed (TODO)" : "Failed";
        $self->_print_to_fh( $self->_diag_fh, "\n" ) if $ENV{HARNESS_ACTIVE};

        my( undef, $file, $line ) = $self->caller;
        if( defined $name ) {
            $self->diag(qq[  $msg test '$name'\n]);
            $self->diag(qq[  at $file line $line.\n]);
        }
        else {
            $self->diag(qq[  $msg test at $file line $line.\n]);
        }
    }

    $self->is_passing(0) unless $test || $self->in_todo;

    # Check that we haven't violated the plan
    $self->_check_is_passing_plan();

    return $test ? 1 : 0;
}

And the second



sub ok {
    my $self = shift;
    my $test = shift;
    my $name = shift;

    $self->assert_start();

    my $result = $self->result_class->new_result(
        description     => $name,
        pass            => $test,
    );

    my $top = $self->top_stack->top;
    $result->file($top->filename);
    $result->line($top->line);

    $self->assert_end($result);

    return $result;
}

And here's Test::Builder2's
Why is it so much simpler?
Partially because its better factored.



sub ok {
    my( $self, $test, $name ) = @_;

    if ( $self->{Child_Name} and not $self->{In_Destroy} ) {
        $name = 'unnamed test' unless defined $name;
        $self->is_passing(0);
        $self->croak("Cannot run test ($name) with active children");
    }
    # $test might contain an object which we don't want to accidentally
    # store, so we turn it into a boolean.
    $test = $test ? 1 : 0;

    lock $self->{Curr_Test};
    $self->{Curr_Test}++;

    # In case $name is a string overloaded object, force it to stringify.
    $self->_unoverload_str( \$name );

    $self->diag(<<"ERR") if defined $name and $name =~ /^[\d\s]+$/;
    You named your test '$name'.  You shouldn't use numbers for your test names.
    Very confusing.
ERR

    # Capture the value of $TODO for the rest of this ok() call
    # so it can more easily be found by other routines.
    my $todo    = $self->todo();
    my $in_todo = $self->in_todo;
    local $self->{Todo} = $todo if $in_todo;

    $self->_unoverload_str( \$todo );

    my $out;
    my $result = &share( {} );

    unless($test) {
        $out .= "not ";
        @$result{ 'ok', 'actual_ok' } = ( ( $self->in_todo ? 1 : 0 ), 0 );
    }
    else {
        @$result{ 'ok', 'actual_ok' } = ( 1, $test );
    }

    $out .= "ok";
    $out .= " $self->{Curr_Test}" if $self->use_numbers;

Most of the work that TB1 is doing is formatting and history



    if( defined $name ) {
        $name =~ s|#|\\#|g;    # # in a name can confuse Test::Harness.
        $out .= " - $name";
        $result->{name} = $name;
    }
    else {
        $result->{name} = '';
    }

    if( $self->in_todo ) {
        $out .= " # TODO $todo";
        $result->{reason} = $todo;
        $result->{type}   = 'todo';
    }
    else {
        $result->{reason} = '';
        $result->{type}   = '';
    }

    $self->{Test_Results}[ $self->{Curr_Test} - 1 ] = $result;
    $out .= "\n";

    $self->_print($out);

    unless($test) {
        my $msg = $self->in_todo ? "Failed (TODO)" : "Failed";
        $self->_print_to_fh( $self->_diag_fh, "\n" ) if $ENV{HARNESS_ACTIVE};

        my( undef, $file, $line ) = $self->caller;
        if( defined $name ) {
            $self->diag(qq[  $msg test '$name'\n]);
            $self->diag(qq[  at $file line $line.\n]);
        }
        else {
            $self->diag(qq[  $msg test at $file line $line.\n]);
        }
    }

    $self->is_passing(0) unless $test || $self->in_todo;

    # Check that we haven't violated the plan
    $self->_check_is_passing_plan();

    return $test ? 1 : 0;
}

More formatting and history



        .-----.                                         .-----.
        | TB2 |                                         | TB1 |
        '-----'                                         '-----'
           |                                               |
           |                                               |
           v                                               v
    .-------------.    .-----------------------.     .-------------.
    | TB2::Result |--->| TB2::EventCoordinator |<----| TB2::Result |
    '-------------'    '-----------------------'     '-------------'
                                   |
                                   v
                          .----------------.
                          | TB2::Formatter |
                          '----------------'
                                   |
                                   v
                              .--------.
                              | Output |
                              '--------'

Here's how TB1 and TB2 relate
They share the same History and Formatter objects
As they run asserts they create Results
    and pass them to History and the Formatter
And the single Formatter object coordinates the output
    it contains the counter (if necessary)
There's also an EventCoordinator but more on that later



Test::Builder
  Test::Builder2::EventCoordinator
    Test::Builder2::History
    Test::Builder2::Formatter

Test::Builder is already using the History and Formatter
  in the 2.0 branch
It was remarkably easy to adapt
It threw out a lot of code



sub ok {
    my( $self, $test, $name ) = @_;

    if ( $self->{Child_Name} and not $self->{In_Destroy} ) {
        $name = 'unnamed test' unless defined $name;
        $self->is_passing(0);
        $self->croak("Cannot run test ($name) with active children");
    }
    # $test might contain an object which we don't want to accidentally
    # store, so we turn it into a boolean.
    $test = $test ? 1 : 0;

#    lock( $self->history );

    # In case $name is a string overloaded object, force it to stringify.
    $self->_unoverload_str( \$name );

    $self->diag(<<"ERR") if defined $name and $name =~ /^[\d\s]+$/;
    You named your test '$name'.  You shouldn't use numbers for your test names.
    Very confusing.
ERR

    # Capture the value of $TODO for the rest of this ok() call
    # so it can more easily be found by other routines.
    my $todo    = $self->todo();
    my $in_todo = $self->in_todo;
    local $self->{Todo} = $todo if $in_todo;
    $self->_unoverload_str( \$todo );

    # Turn the test into a Result
    my( $pack, $file, $line ) = $self->caller;
    my $result = Test::Builder2::Result->new_result(
        pass            => $test ? 1 : 0,
        location        => $file,
        id              => $line,
        description     => $name,
        directives      => $in_todo ? ["todo"] : [],
        reason          => $in_todo ? $todo : undef,
    );

    # Store the Result in history making sure to make it thread safe
    $result = shared_clone($result);
    $self->stream_start unless $self->stream_started;
    $self->event_coordinator->post_result($result);

    $self->is_passing(0) unless $test || $self->in_todo;

    # Check that we haven't violated the plan
    $self->_check_is_passing_plan();

    $self->last_test_seen( $self->current_test );

    return $test ? 1 : 0;
}

Here's Test::Builder->ok using History and Formatter
Not quite half the size
But certainly far simpler



GOODGOOD
This is a good sign
It indicates the design is as flexible as I'd hoped
And if this whole thing fails
TB1 can get the history and formatter



Test::Builder
Test::Builder2
Fennic ?
Test::Assert ?

What's happened is instead of having one test builder object
We now have many
With different APIs
And different ideas about how asserts work
Coordinated by a handful of simpler, lower level objects



Its a federated system of test builders
This is VERY different from where I started
What it means is my ideas and prejudices about how tests are written
   can be ignored
Don't like how Test::Builder2 works?
  and I'm a little worried about it
Write your own!  Please.
The Formatter and History objects are very simple.
And impose few restrictions.



Result

History

Formatter

EventCoordinator

Sharing these pieces you can write your own builder
I intend to eventually push them into their own release
    separate from Test::Builder2 and Test::More
To put everyone on an equal footing



Test::Builder2::Formatter

I want to show off the formatter a little bit
Because its so simple to make a new one
And I'd really love to see a GUI formatter



But first we have to talk about events
And Event Watchers
Which let multiple builders all coordinate
And let 3rd party testing modules affect them
Everything of interest in a test generates an event.



use Test::More tests => 2;

ok 1, "true is true";
is 1+1+1, 2, "addition";

This simple test



stream_start
set_plan
    asserts_expected => 2
result
    pass => 1
    name => "true is true"
result
    pass => 0
    name => "addition"
stream_end

Generates these events



And they get passed around by the EventCoordinator
To EventWatchers
TAP Formatter is one



The EventCoordinator was the last major design problem
rjbs and Dieter solved it over drinks at PPW 2010



1..2
ok 1 - true is true
not ok 2 - addition
#   Failed test 'addition'
#   at example.t line 4.
# 1 test of 2 failed.

And it takes those events and spits out TAP
The formatter is completely separate from the builder
So they can be very flexible



# full code in examples/TB2/

package TB2::Formatter::SimpleHTML;

use Test::Builder2::Mouse;
extends "Test::Builder2::Formatter";

An example of a very simple HTML formatter
I'm going to skip some details
The full code is in examples
Start by inheriting from the base formatter



sub accept_stream_start {
    my $self = shift;

    $self->write(out => <<"HTML");
<html>
<head>
  <title>TB2::Formatter::SimpleHTML demo</title>
</head>
<body>
<table>
  <tr><th>Result</th><th>Name</th></tr>
HTML

    return;
}

Define accept_stream_start
This means testing has begun
Spit out the HTML header and a table to hold test results
It has a "streamer" to accept the output via write()
The streamer can be swapped out



sub accept_result {
    my $self = shift;
    my $result = shift;

    my $name = $result->name || '';
    my $ok   = $result ? "pass"
                       : "<b>fail</b>";
    $self->write(out => <<"HTML");
    <tr><td>$ok</td><td>$name</td></tr>
HTML

    return;
}

Formatting a result object.



sub accept_stream_end {
    my $self = shift;

    $self->write(out => <<"HTML");
</table>
</body>
</html>
HTML

    return;
}

And when you're told testing is done, spit out the end tags.



my $tb2 = Test::Builder2->singleton;
my $ec  = $tb2->event_coordinator;

$ec->clear_formatters;
$ec->add_formatters(
    TB2::Formatter::SimpleHTML->create
);

Then you can plug that into the event coordinator
(sorry, it's a bit inconvenient right now)



use Test::Simple tests => 2;
ok(1, "this passed");
ok(0, "this failed");

And run some tests against it



<html>
<head>
  <title>TB2::Formatter::SimpleHTML demo</title>
</head>
<body>
<table>
    <tr><th>Result</th><th>Name</th></tr>
    <tr><td>pass</td><td>this passed</td></tr>
    <tr><td><b>fail</b></td><td>this failed</td></tr>
</table>
</body>
</html>

Which produces HTML as output instead of TAP



And a really boring HTML page



I can make it more interesting
But I hope you get the idea
The point is the formatter is almost completely decoupled from everything else.
So you can DO almost anything you want in a Formatter.
  And you can hook what you want to a formatter.
And a formatter can ignore as much of the events as it wants
Like the plan



Test::Builder2::Formatter::Null

There's the null formatter which does nothing



Test::Builder2::Formatter::Multi

There's the multiplexer which takes other formatters
  and passes the results through to them all
Why would you do that?



Test::Builder2::Formatter::Multi
    Formatter::GreenBar
    Formatter::TAP

Maybe you want graphical output
And still output TAP



Test::Builder2::Formatter::Multi
    Formatter::TAP
    Formatter::JUnit
        Streamer::ToHudson

Or maybe you want to output TAP
And send JUnit XML to your Hudson continuous testing server

Similarly, there's the History object
It stores a record of all the results and events
And keeps statistics
It's mostly used to spit out end of test information



my $history = $tb->history;

say $history->is_passing
      ? "PASS"
      : "FAIL"
      ;

say sprintf "%d tests skipped",
    $history->skip_count;

my $results = $history->results;

And yes, there's a version which stores no results
For people who run lots and lots of tests in a single process
But it does still keep statistics
Like the formatter, history just takes results and events.
  its completely decoupled



Die on fail.



package TB2::DieOnFail::Watcher;

use Test::Builder2::Mouse;
with 'Test::Builder2::EventWatcher';

sub accept_event {}

sub accept_result {
    my $self   = shift;
    my $result = shift;

    return if $result;

    my $name = $result->name;
    die sprintf "Test%s failed, aborting.\n",
        defined $name ? " $name" : "";
};

Here's die-on-fail
Implemented as an event watcher



my $w = TB2::DieOnFail::Watcher->new;

Test::Builder2
    ->singleton
    ->event_coordinator
    ->add_late_watchers(
        $w
    );

And then you add the watcher to the event coordinator
And it will see everything



package TB2::DebugOnFail::Watcher;

use Test::Builder2::Mouse;
with 'Test::Builder2::EventWatcher';

sub accept_event {}

sub accept_result {
    my $self   = shift;
    my $result = shift;

    return if $result;

    $DB::single = 1;
    return;  # welcome to the debugger.
             # $result contains the result
};

Debug on fail is more interesting to me



use TB2::DebugOnFail;
use Test::Simple tests => 3;

ok(1, "this passes");
ok(0, "this fails");
ok(1, "and another pass");



$ perl -dw ~/tmp/test.plx

Loading DB routines from perl5db.pl version 1.33
Editor support available.

Enter h or `h h' for help, or `man perldebug' for more help.

TAP version 13
1..3
main::(/Users/schwern/tmp/test.plx:7):
7: ok(1, "this passes");
  DB<1> c                                                                                           
ok 1 - this passes
not ok 2 - this fails
#   Failed test 'this fails'
#   at /Users/schwern/tmp/test.plx line 8.
TB2::DebugOnFail::Watcher::accept_result(examples/TB2/lib/TB2/
DebugOnFail.pm:56):
56:         return;  # welcome to the debugger.  $result contains the 
result
  DB<1> x $result->file                                                                             
0  '/Users/schwern/tmp/test.plx'
  DB<2> x $result->line                                                                             
0  8

It dumps you in the middle of DebugOnFail
Its enough to prove it works
A debugger guru can tell me if there's a way to break
   on a specific line and file



package TB2::NoWarnings::WarningsWatcher;

use Test::Builder2::Mouse;
with 'Test::Builder2::EventWatcher';

has builder  =>
  is                 => 'rw',
  isa                => 'Test::Builder2',
  default            => sub {
      require Test::Builder2;
      return Test::Builder2->singleton;
  };

has warnings_seen =>
  is                 => 'rw',
  isa                => 'ArrayRef',
  default            => sub { [] };

has quiet_warnings =>
  is                 => 'rw',
  isa                => 'Bool',
  default            => 0;

Test::NoWarnings is the most complicated real use case
  and it works
Start by setting up some accessors using Mouse



sub accept_stream_start {
    my $self = shift;

    $SIG{__WARN__} = sub {
        push @{$self->warnings_seen}, @_;
        warn @_ unless $self->quiet_warnings;
    };

    return;
}

When we see an event telling us testing has started
Set up our warning handler to capture warnings



sub accept_set_plan {
    my $self  = shift;
    my $event = shift;

    my $want = $event->asserts_expected;
    $event->asserts_expected( $want + 1 )
      if $want;

    return;
}

If someone sets the plan with a number
Add one for our own test



sub accept_stream_end {
    my $self = shift;

    my $warnings = $self->warnings_seen;
    my $tb = $self->builder;

    my $result = $tb->ok(
        scalar @$warnings,
        "no warnings"
    );

    $result->diagnostic([
        warnings => $warnings
    ]);

    remove_warning_handler();

    return;
}

Finally, just before testing is over,
run our own test
and remove our warning handler



package TB2::NoWarnings;

sub import {
    no_warnings();
}

sub no_warnings {
    my $watcher = 
      TB2::NoWarnings::WarningsWatcher->new(
        @_
      );

    $watcher->builder
            ->event_coordinator
            ->add_early_watchers($watcher);

    return;
}

When you load TB2::NoWarnings
It adds its watcher as an *early* watcher
So it can alter the set_plan event
And get in a last test before the formatter finishes



examples/TB2/

These are in the repository
They're decently documented



What's the future?
Alpha 2 was the end of the design phase.
The scope has stopped expanding.
All the major pieces are working.
This is the equivalent of Larry saying
  "the Perl 6 language is basically done"
  "we're going to finish the implementation now"
Alpha releases will start to become more regular



The interface is still really rough, needs improved
  Didn't want to write a bunch of convenience just to tear it out
Convert Test::More
  Bring assert methods to TB2
Complete TB1 conversion
Test against CPAN



Convert existing Test modules
  (I'll happily help)
Sexy formatters
Test::Builder2::Tester (implemented as a watcher)
Getting more people involved



How can you help?
I don't get a lot of help.
Test::Builder2 isn't very convenient
And its at least two layers below where most people care about.



Test::Builder2::Design

There's a design document here
It contains what I said in detail
It also contains the special considerations of TB2
Like...



It has to work
And it has to work in every environment
Because every CPAN that matters relies on it
Which severely restrains the design



http://github.com/schwern
       /test-more/issues/

Click: Test-Builder2
Click: Easy

I try to keep the current tasks up to date in the issue tracker
The TB2 ones are in their label
And there's easy tasks



Smoke It

If you run a smoker, load it with the latest 2.00 alpha
    or even from the repo
Compare against the latest stable
Report the differences in the issue tracker



Write a
Formatter



Write a
Formatter

I'm not good at making things sparkly
Hudson, JUnit, HTML, XML, GUI, POSIX, audio?
Give us feedback about what was easy
  what was hard
  what was impossible



Port
Your
Test

Module
For the brave
A lot of things are missing



Test
Against
Your

Module
CPAN is the best test suite
Try it against your code
Report if it fails



PERL5LIB=/path/to/TB2/lib make test

You don't have to install it
(please don't install it)
You can just download it and do a one off



There's a bunch of things that are known to be broken



threads
subtest()
Test::Aggregate
Test::Group
Test::Class
Test::SharedFork

These are all busted
Test::Class is easily fixed
The rest, dunno yet



•rjbs for designing the Formatter
•rjbs + hdp for drunken EventCoordinator design
•notbenh for History work
•chromatic for writing TB1
•sartak & the Mouse folks
•PDX Hackathon for listening to me bitch
•Ovid for bitching about die-on-fail
•Alberto Simões & TPF for being very patient
•Andreas König for smoke testing
•TPF contributors for your money
•any contributors and module authors I forgot

Thanks!



Thanks!



Questions?


